
This is a GAUSS procedure.
It computes a test for a threshold in linear regression
allowing for heteroskedasticity.  The procedure takes the form

{f_test,p_value} = thr_test(dat,yi,xi,qi,trim_per,rep)

The inputs are:
dat      = data matrix (nxk)
yi       = index of dependent (y) variable, e.g. yi = 1
xi       = indexes of independent (x) variables, e.g. xi = 2|3
qi       = index of threshold (q) variable, e.g. qi = 4
trim_per = percentage of sample to trim from ends, e.g. trim_per = .15
rep      = number of bootstrap replications, e.g. rep = 1000

Output:
f_test   = Value of Maximal (Quandt) F-statistic
p_value  = Bootstrap p-value

There is one global variable:

_graph   = 0;     @ Graph indicator
                    Set _graph=0 to not view the graph of the likelihood
                    Set _graph=1 to view the graph of the likelihood and print output @

Notes:
(1)    Do not include a constant in the independent variables,
       the program automatically adds an intercept to the regression.

(2)    There are two bootstrap methods which the program can use.

       The first method, obtained by setting _quick=1 at the beginning
       of the procedure code, is the method presented in my paper
       "Inference When a Nuisance Parameter is Not Identified Under
       the Null Hypothesis" which simulates the asymptotic null
       distribution.  A computational shortcut is also taken which speeds
       computational time, at the cost of greater memory usage, so may
       not be possible for large data sets.

       The second method, obtained by setting _quick=2 at the beginning
       of the procedure code, is a "fixed regressor bootstrap", which is
       quite close.  The difference is that the bootstrap procedure calculates
       the variance-covariance matrix in each bootstrap replication.  This
       results in a better finite sample approximation.  The cost is greater
       computation time.

       The program is set by default to use the second method, which has better
       sampling properties.  If computational time is a concern, switch to the
       first (set _quick=1).  If an "out of workspace memory" message appears,
       switch back to _quick=2.


Example:
If the nxk matrix "dat" contains the dependent variable in the first
column, the independent variables in the second through tenth columns,
and the threshold variable in the fifth:

xi = 2|3|4|5|6|7|8|9|10;
{f_test,p_value} = thr_test(dat,1,xi,5,.15,1000);


*/

/*******************************************************************/

proc (2) = thr_het(dat,yi,xi,qi,trim_per,rep);
local n,q,y,x,k,mi,e,ee,xe,qs,qn,sn,mm,sume,mmistore,r,
smax,sig,ftest,fboot,j,pv,cr,lr,clr,mmi,qmax,si,qq,cqq,cqqr,
cqqb,sq,tit,_cr,_quick,cxe,vv,vi,u,sm;


/******************************************************
Control Parameters, may be changed  */

_cr = .95;        @ This is the confidence level used to plot the
                    critical value in the graph.  It is not used
                    elsewhere in the analysis.                          @
_quick   = 2;     @ Indicator of method used for bootstrap
                    Set _quick=1 for quick computation of asymptotic
                    distribution.  This method is not a proper bootstrap
                    and may result in excess rejections.  It also uses
                    more memory.
                    Set _quick=2 for a better bootstrap procedure, which
                    also uses less memory, but is more time consuming   @


/******************************************************/

n  = rows(dat);
q = dat[.,qi];
qs = sortind(q);
y = dat[qs,yi];
x = ones(n,1)~dat[qs,xi];
q = q[qs];
k = cols(x);
qs = unique(q,1);
qn = rows(qs);
qq = zeros(qn,1);
r=1; do while r<=qn;
  qq[r] = sumc(q.==qs[r]);
r=r+1; endo;
cqq = cumsumc(qq);
sq = (cqq.>=floor(n*trim_per)).*(cqq.<=(floor(n*(1-trim_per))));
qs = selif(qs,sq);
cqq = selif(cqq,sq);
qn = rows(qs);

mi = invpd(moment(x,0));
e  = y-x*mi*(x'y);
ee = e'e;
xe = x.*e;
vi = moment(xe,0);
cxe = cumsumc(xe);
sn = zeros(qn,1);

if _quick == 1;
  mmistore = zeros(k*(k+1)/2,qn);
  cqqb = 1;
  mm = zeros(k,k);
  vv = zeros(k,k);
  r=1; do while r<=qn;
    cqqr = cqq[r];
    mm = mm + moment(x[cqqb:cqqr,.],0);
    vv = vv + moment(xe[cqqb:cqqr,.],0);
    sume = cxe[cqqr,.]';
    mmi = invpd(vv - mm*mi*vv - vv*mi*mm + mm*mi*vi*mi*mm);
    sn[r] = sume'mmi*sume;
    cqqb = cqqr+1;
    mmistore[.,r] = vech(mmi);
  r=r+1;endo;
  si = maxindc(sn);
  qmax = qs[si];
  lr = sn;
  ftest = sn[si];
  fboot = zeros(rep,1);
  j=1; do while j<=rep;
    y  = rndn(n,1).*e;
    xe = x.*(y-x*mi*(x'y));
    cxe = cumsumc(xe);
    sn = zeros(qn,1);
    r=1; do while r<=qn;
      mmi = xpnd(mmistore[.,r]);
      sume = cxe[cqq[r],.]';
      sn[r] = sume'mmi*sume;
    r=r+1;endo;
    fboot[j] = maxc(sn);
  j=j+1;endo;
elseif _quick==2;
  cqqb = 1;
  mm = zeros(k,k);
  vv = zeros(k,k);
  r=1; do while r<=qn;
    cqqr = cqq[r];
    mm = mm + moment(x[cqqb:cqqr,.],0);
    vv = vv + moment(xe[cqqb:cqqr,.],0);
    sume = cxe[cqqr,.]';
    mmi = vv - mm*mi*vv - vv*mi*mm + mm*mi*vi*mi*mm;
    trap 1;
    sm = sume/mmi;
    trap 0;
    if scalerr(sm)==0;
      sn[r] = sume'sm;
    endif;
    cqqb = cqqr+1;
  r=r+1;endo;
  si = maxindc(sn);
  qmax = qs[si];
  lr = sn;
  ftest = sn[si];
  fboot = zeros(rep,1);
  j=1; do while j<=rep;
    y  = rndn(n,1).*e;
    xe = x.*(y-x*mi*(x'y));
    vi = moment(xe,0);
    cxe = cumsumc(xe);
    sn = zeros(qn,1);
    cqqb = 1;
    mm = zeros(k,k);
    vv = zeros(k,k);
    r=1; do while r<=qn;
      cqqr = cqq[r];
      mm = mm + moment(x[cqqb:cqqr,.],0);
      vv = vv + moment(xe[cqqb:cqqr,.],0);
      mmi = vv - mm*mi*vv - vv*mi*mm + mm*mi*vi*mi*mm;
      sume = cxe[cqqr,.]';
      trap 1;
        sm = sume/mmi;
      trap 0;
      if scalerr(sm)==0;
        sn[r] = sume'sm;
      endif;
      cqqb = cqqr+1;
    r=r+1;endo;
    fboot[j] = maxc(sn);
  j=j+1;endo;
endif;
fboot = sortc(fboot,1);
pv = meanc(fboot .>= ftest);
cr = fboot[round(rep*_cr)];

if _graph==1;
 _pdate = "";
 fonts("complex simgrma");
 tit = "F Test For Threshold\L";
 tit = tit $+ "Reject Linearity if F Sequence Exceeds Critical Value";
 title(tit);
 ylabel("F]n[(\202g\201)");
 xlabel("\202g\201");
 clr = ones(qn,1)*cr;
 _plegstr="" $+ "LR]n[(\202g\201)\000" $+ ftocv(_cr*100,2,0)
 $+ "% Critical";
 _plegctl = 1;
 xy(qs,lr~clr);
"";"";
"Test of Null of No Threshold Against Alternative of Threshold";
"Allowing Heteroskedastic Errors (White Corrected)";
"";
"Number of Bootstrap Replications " rep;
"Trimming Percentage              " trim_per;
"";
"Threshold Estimate               " qmax;
"LM-test for no threshold         " ftest;
"Bootstrap P-Value                " pv;
"";"";
endif;

retp(ftest,pv);
endp;

